
//   Явная схема для уравнения теплопроводности на отрезке 
 
// Создание сетки:---------------------------------------------------------- 
N=11; // число узлов по x 
// число узлов по t должно удовлетворять условию 
//         M>= 1+2*(N-1)^2: 
M=10+2*round((N-1)^2);  
pi=3.14159;                        
x=zeros(N,1); 
h=1/(N-1); 
for i=1:N 
    x(i)=(i-1)*h; 
end 
 
t=zeros(M,1); 
tau=1/(M-1); 
for j=1:M 
    t(j)=(j-1)*tau; 
end 
//-------------------------------------------------------------------------------- 
 
// численное решение:------------------------------------------------------- 
y=zeros(M,N); 
// начальное условие: 
for i=1:N 
    y(1,i)=sin(3*pi*x(i)/2); 
end 
 
for j=1:M-1 
    // решение уравнения на внутренних узлах сетки: 
    for i=2:N-1 
        y(j+1,i)=y(j,i)+tau*(y(j,i+1)-2*y(j,i)+y(j,i-1))/(h*h)+tau*x(i); 
    end 
    // граничные условия: 
    y(j+1,1)=0;   // в данном случае не обязательно, так как y изначально  
                         // заполнено нулями 
    // первый вариант граничного условия при x=1:             
    //     y(j+1,N)=y(j+1,N-1)+h*t(j+1); 
    // второй вариант граничного условия при x=1:  
     y(j+1,N)=(y(j+1,N-1)+h*t(j+1)+h^2*y(j,N)/(2*tau)+h^2/2)/(1+h^2/(2*tau)); 
    // третий вариант граничного условия при x=1:  
    //    y(j+1,N)=4*y(j+1,N-1)/3-y(j+1,N-2)/3+2*h*t(j+1)/3; 
end 
//----------------------------------------------------------------------------------- 
 



// аналитическое решение и вычисление погрешности:-------------------------- 
u=zeros(M,N);     // аналитическое  решение 
err=zeros(M,N);   // погрешность численного  решения 
 
for i=1:N 
    for j=1:M 
        u(j,i)=x(i)*t(j)+exp(-9*pi*pi*t(j)/4)*sin(3*pi*x(i)/2); 
        err(j,i)=u(j,i)-y(j,i); 
    end 
end 
//----------------------------------------------------------------------------------------------- 
figure 
surf(x,t,y) 
xlabel('x') 
ylabel('t') 
zlabel('y') 
title('numerical solution') 
 
figure 
surf(x,t,u) 
xlabel('x') 
ylabel('t') 
zlabel('u') 
title('analitical solution') 
 
figure 
surf(x,t,err) 
xlabel('x') 
ylabel('t') 
zlabel('err') 
title('error') 
 
 
 
 
 
 
 
 
 
 
 
 
 
 



//   Чисто неявная схема для уравнения теплопроводности на отрезке 
 
// Создание сетки:------------------------------------------------------------------ 
N=51; // число узлов по x 
M=51; // число узлов по t  
pi=3.14159;                        
x=zeros(N,1); 
h=1/(N-1); 
for i=1:N 
    x(i)=(i-1)*h; 
end 
 
t=zeros(M,1); 
tau=1/(M-1); 
for j=1:M 
    t(j)=(j-1)*tau; 
end 
//---------------------------------------------------------------------------------------- 
 
// численное решение:-------------------------------------------------------------- 
y=zeros(M,N); 
// начальное условие: 
for i=1:N 
    y(1,i)=sin(3*pi*x(i)/2); 
end 
// Создание массивов прогоночных коэффициентов: 
alpha=zeros(N-1,1); 
bet=zeros(N-1,1); 
// Параметры системы с трехдиагональной матрицей: 
A=tau/(h^2); 
B=A; 
C=2*A+1; 
 
for j=1:M-1 
    // прямой ход прогонки: 
    // прямой ход прогонки:---------------------------------------------------------- 
    // alpha(1)= 
    // bet(1)= 
    // в данном примере alpha(1)=0 и bet(1)=0 
 
    for i=2:N-1 
        F=y(j,i)+tau*x(i); 
        alpha(i)=B/(C-A*alpha(i-1)); 
        bet(i)=(A*bet(i-1)+F)/(C-A*alpha(i-1)); 
    end 



    // первый вариант граничного условия при x=1: 
    //     kappa_2=1; 
    //     mu_2=h*t(j+1); 
    // второй вариант граничного условия при x=1: 
     kappa_2=2*tau/(2*tau+h^2); 
     mu_2=kappa_2*(h*t(j+1)+(h^2)*(1+y(j,N)/tau)/2); 
    // третий вариант граничного условия при x=1: 
    //    kappa_2=1-(h^2)/(2*tau); 
    //    mu_2=(h^2)*F/(2*tau)+h*t(j+1); 
    // обратный ход прогонки: 
    y(j+1,N)=(mu_2+kappa_2*bet(N-1))/(1-kappa_2*alpha(N-1)); 
    for i=N-1:-1:1 
        y(j+1,i)=alpha(i)*y(j+1,i+1)+bet(i); 
    end 
end 
// аналитическое решение и вычисление погрешности:-------------------------- 
u=zeros(M,N); 
err=zeros(M,N); 
for i=1:N 
    for j=1:M 
        u(j,i)=x(i)*t(j)+exp(-9*pi*pi*t(j)/4)*sin(3*pi*x(i)/2); 
        err(j,i)=u(j,i)-y(j,i); 
    end 
end 
//----------------------------------------------------------------------------------------------- 
figure 
surf(x,t,y) 
xlabel('x') 
ylabel('t') 
zlabel('y') 
title('numerical solution') 
 
figure 
surf(x,t,u) 
xlabel('x') 
ylabel('t') 
zlabel('u') 
title('analitical solution') 
 
figure 
surf(x,t,err) 
xlabel('x') 
ylabel('t') 
zlabel('err') 
title('error') 



//   Симметричная неявная схема для уравнения теплопроводности на  
//  отрезке 
 
// Создание сетки:---------------------------------------------------------- 
N=51; // число узлов по x 
M=51; // число узлов по t  
pi=3.14159;                        
x=zeros(N,1); 
h=1/(N-1); 
for i=1:N 
    x(i)=(i-1)*h; 
end 
 
t=zeros(M,1); 
tau=1/(M-1); 
for j=1:M 
    t(j)=(j-1)*tau; 
end 
//--------------------------------------------------------------------------------- 
 
// численное решение:------------------------------------------------------- 
y=zeros(M,N); 
// начальное условие: 
for i=1:N 
    y(1,i)=sin(3*pi*x(i)/2); 
end 
// Создание массивов прогоночных коэффициентов,  
// заполненных нулями: 
alpha=zeros(N-1,1); 
bet=zeros(N-1,1); 
// Параметры системы с трехдиагональной матрицей: 
A=tau/(2*h^2); 
B=A; 
C=2*A+1; 
 
for j=1:M-1 
    // прямой ход прогонки:--------------------------------------------------- 
     // alpha(1)= 
    // bet(1)= 
    // в данном примере alpha(1)=0 и bet(1)=0 
    for i=2:N-1 
        F=y(j,i)+tau*x(i)+tau*(y(j,i-1)-2*y(j,i)+y(j,i+1))/(2*h^2); 
        alpha(i)=B/(C-A*alpha(i-1)); 
        bet(i)=(A*bet(i-1)+F)/(C-A*alpha(i-1)); 
    end 



    // первый вариант граничного условия при x=1: 
    // kappa_2=1; 
    // mu_2=h*t(j+1); 
    // второй вариант граничного условия при x=1: 
    kappa_2=1/(1+h^2/tau); 
    mu_2=kappa_2*(2*h*(t(j)+tau/2)+(1+y(j,N)/tau)*h^2 -y(j,N)+y(j,N-1)); 
    // третий вариант граничного условия при x=1: 
    //  kappa_2=1-h^2/tau; 
    //  mu_2=h^2*F/tau+h*t(j+1); 
    // обратный ход прогонки:----------------------------------------------------------- 
    y(j+1,N)=(mu_2+kappa_2*bet(N-1))/(1-kappa_2*alpha(N-1)); 
    for i=N-1:-1:1 
        y(j+1,i)=alpha(i)*y(j+1,i+1)+bet(i); 
    end 
end 
// аналитическое решение и вычисление погрешности:-------------------------- 
u=zeros(M,N);    // аналитическое  решение 
err=zeros(M,N);  // погрешность численного  решения 
for i=1:N 
    for j=1:M 
        u(j,i)=x(i)*t(j)+exp(-9*pi*pi*t(j)/4)*sin(3*pi*x(i)/2); 
        err(j,i)=u(j,i)-y(j,i); 
    end 
end 
//----------------------------------------------------------------------------------------------- 
figure 
surf(x,t,y) 
xlabel('x') 
ylabel('t') 
zlabel('y') 
title('numerical solution') 
 
figure 
surf(x,t,u) 
xlabel('x') 
ylabel('t') 
zlabel('u') 
title('analitical solution') 
 
figure 
surf(x,t,err) 
xlabel('x') 
ylabel('t') 
zlabel('err') 
title('error') 


